ЛАБОРАТОРНАЯ РАБОТА 3

Математическое моделирование

Скандарова Полина Юрьевна

Цель работы

- Построить графики модели боевых действий, а также ознакомиться с Scilab.

Вариант 26

Задача: Между страной X и страной У идет война. Численность состава войск исчисляется от начала войны, и являются временными функциями x(t) и y(t). В начальный момент времени страна X имеет армию численностью 80 000 человек, а в распоряжении страны У армия численностью в 115 000 человек. Для упрощения модели считаем, что коэффициенты a,b,c,h постоянны. постоянны. Также считаем P(t) и Q(t) непрерывные функции. Постройте графики изменения численности войск армии X и армии У для следующих случаев:

1. Модель боевых действий между регулярными войсками

$$\frac{\partial x}{\partial t} = -0{,}3x(t) - 0{,}56y(t) + sin(t + 10)$$
$$\frac{\partial y}{\partial t} = -0{,}68x(t) - 0{,}33y(t) + cos(t + 10)$$

2. Модель ведение боевых действий с участием регулярных войск и партизанских отрядов

$$\frac{\partial x}{\partial t} = -0{,}31x(t) - 0{,}77y(t) + sin(2t + 10)$$
$$\frac{\partial y}{\partial t} = -0{,}67x(t)y(t) - 0{,}51y(t) + cos(t + 10)$$

1. Рассмотрим подробнее уравнения

1.1. Потери, не связанные с боевыми действиями, описывают члены -0,3x(t) и -0,33y(t), члены -0,56y(t) и -0,68x(t) отражают потери на поле боя. Функции P(t)=sin(t+10), Q(t)=cos(t+10) учитывают возможность подхода подкрепления к войскам X и У в течение одного дня.

1.2. Потери, не связанные с боевыми действиями, описывают члены -0,31x(t) и -0,51y(t), члены -0,77y(t) и -0,67x(t)y(t) отражают потери на поле боя. Функции P(t)=sin(2t+10), Q(t)=cos(t+10) учитывают возможность подхода подкрепления к войскам X и У в течение одного дня.

1.3. Начальные условия для обоих случаев будут равно $x_0 = 80000$, $y_0 = 115000$

2. Построение графиков численности войск

2.1. Напишем первую программу для Scilab:

```
//начальные условия
x0 = 80000;//численность первой армии
y0 = 115000;//численность второй армии
t0 = 0;//начальный момент времени
a = 0.3;//константа, характеризующая степень влияния различных факторов на по
тери
b = 0.56;//эффективность боевых действий армии y
c = 0.68;//эффективность боевых действий армии x
d = 0.33;//константа, характеризующая степень влияния различных факторов на п
отери
tmax = 1;//предельный момент времени
dt = 0.05;//шаг изменения времени
t = [t0:dt:tmax];
function p = P(t)//возможность подхода подкрепления к армии x
p = sin(t+10);
endfunction
function q = Q(t)//возможность подхода подкрепления к армии y
q = cos(t+10);
endfunction
//Система дифференциальных уравнений
function dy = syst(t, y)
dy(1) = - a*y(1) - b*y(2) + P(t);//изменение численности первой армии
dy(2) = - c*y(1) - d*y(2) + Q(t);//изменение численности второй армии
endfunction
v0 = [x0;y0];//Вектор начальных условий
//Решение системы
y = ode(v0,t0,t,syst);
//Построение графиков решений
scf(0);
plot2d(t,y(1,:),style=2);//График изменения численности армии x(синий)
xtitle("CombatModel#1","Step","ArmySize");
plot2d(t,y(2,:), style = 5);//График изменения численности армии y (красный)
xgrid();
```



Рис. 1: График для первого случая

2.2. Напишем вторую программу для Scilab:

```
x0 = 80000;
y0 = 115000;
t0 = 0;
a = 0.31;
b = 0.77;
c = 0.67;
d = 0.51;
tmax = 1;
dt = 0.05;
t = [t0:dt:tmax];
function p = P(t)
p = sin(2*t+10);
endfunction
function q = Q(t)
q = cos(t+10);
endfunction
//Система дифференциальных уравнений
function dy = syst(t, y)
dy(1) = - a*y(1) - b*y(2) + P(t);
dy(2) = - c*y(1)*y(2) - d*y(2) + Q(t);
endfunction
v0 = [x0;y0];
y = ode(v0,t0,t,syst);
scf(0);
plot2d(t,y(1,:),style=2);
xtitle("CombatModel#2","Step","ArmySize");
plot2d(t,y(2,:), style = 5);
xgrid();
```



Рис. 2: График для второго случая

Вывод

- В результате выполнения лабораторной работы мы научились решать и строить графики модели боевых действий в среде Scilab.